

The Applications of Tolerance Intervals: Make it Easy

Marina Komaroff, Noven Pharmaceuticals, Inc., New York, NY

ABSTRACT

It is known that Confidence Intervals (CI) cover a certain population parameter with a stated confidence; however, there is also a way to cover a *fixed proportion of the population* for a characteristic of interest with a specified confidence and such interval is named Tolerance Interval (TI).

The application of TI(s) in clinical trials has been used for defining drug response with biomarker(s). Transdermal medicine that requires evaluation of stability and drug release to adhesive transfer for the evaluation quality of transdermal patches relies on calculation of TIs. SAS Institute provides a way to calculate TIs with PROC CAPABILITY but requires the SAS/QC package. However, most users do not need nor can afford this entire package for industrial engineering when the need for tolerance intervals can be rare and unpredictable. That means that researchers need to write their own SAS code.

The author is convinced that easy calculation may potentially expand the use of TIs in different applications of clinical trials. This paper explains the algorithm involved in TI calculation, and provides a user-friendly SAS MACRO that uses only SAS/BASE. Statistical analysis was conducted using the SAS System for Windows (release 9.3; SAS Institute Inc., Cary, N.C.)

INTRODUCTION

Confidence Intervals (CIs) are intervals that cover population parameter (such as mean, median) with a certain confidence. The formula to calculate CI:

$$CI = Mean \pm \left(\frac{1}{n}\right)^{0.5} * t_{1-\frac{\alpha}{2}, n-1} * StandardDeviation$$

If parameter has Binomial distribution with probability of event $p=0.5$, then: $CI = n * p \pm z\sqrt{n * p(1-p)}$, where z depends upon the confidence level. For example, what is 95% CI for the median of 100 observations?

$$CI_{Lower} = 100*(.5) - 1.96*sqrt(100*.5*(1-.5)) = 40$$

$$CI_{Upper} = 100*(.5) + 1.96*sqrt(100*.5*(1-.5)) = 60$$

Answer: The 95% CI for median is between the 40th to the 60th observation.

Prediction Intervals (PIs) are intervals that cover m future individual values. The formula to calculate PI:

$$PI = Mean \pm \left(1 + \frac{1}{n}\right)^{0.5} * t_{1-\frac{\alpha}{2*m}, n-1} * StandardDeviation$$

Compared to CI, PIs are wider because they take into account the variability of individual data points around the summary statistics like means or medians. Non-parametric PIs are computed as following:

$PI_{Lower} = (n + 1) * \frac{\alpha}{2}$ and $PI_{Upper} = (n + 1) * \left(1 - \frac{\alpha}{2}\right)$, where $(1-\alpha)$ represents the level of confidence. For example, what is 95% PI for the 100 observations?

$$PI_{Lower} = 101*(.05/2) = 3$$

$$PI_{Upper} = 101*(1-.05/2) = 99$$

Answer: The 95% PI ranges from the 3rd to the 99th observation.

If data is given, then PI can be calculated with $100*(n-1)/(n+1)$ % confidence that the future observation will fall between the range of observed data. For example, data for 150 observations has the range (min=85, max=91). There is $100*(150-1)/(150+1)=98.7\%$ confidence that the next observation will be

within 85 and 91.

Tolerance Intervals (TIs) show the limits for *proportions of future individual values with (1-α) level of confidence*. Compared to PI, TI covers proportion of individual data within a range versus a single future observation. In other words, there is the shortest tolerance interval [A B] that covers p proportion of data with some probability (confidence). If p (proportion) or (1-α) confidence increases than the interval [A B] gets wider.

If parameter has normal distribution, then two-sided: $TI = SampleMean \pm KFactor * StandardDeviation$, where KFactor is function of the confidence level (1-α), the proportion p, and the number of observations n. The formula to calculate 2-sided TI (Howe, 1969):

$$TI_{2sd} = Mean \pm \sqrt{\frac{(n-1)(1+\frac{1}{n})Z_{(1+p)/2}^2}{\chi_{(\alpha, n-1)}^2}} * StandardDeviation$$

One-sided TI: $TI_{1s} = Mean \pm \frac{z_p + \sqrt{z_p^2 - A*B}}{A} * StandardDeviation$; where [A,B] is the shortest interval that covers p-proportion of data with (1-α) confidence.

If data is not normally distributed, than Non-Parametric TI (NPTI) should be applied, (Turkey, 1947). Two-sided NPTI = [Y_k, Y_{n-k+1}] where Y_k is the kth value, and Y_{n-k+1} is the n-k+1 value of the ordered parameter of interest Y. And k is the largest integer that satisfies the criteria: $CDF Beta_{n-2k+1, 2k}(p) \leq \alpha$, where CDF Beta is cumulative distribution function of beta distribution.

One-sided NPTI=[Y_k, Y_{n-k}] where k is the largest integer that satisfies: $CDF Beta_{n-k+1, k}(p) \leq \alpha$.

APPLICATIONS

There are multiple areas where Tolerance Intervals play important role. Some of them are listed below.

1. Quality control of drug products
 - a. TIs can be applied to predict future outcomes for some production process, and establish criteria for quality metrics data.^[1,2] For example, TIs help the development and revision of acceptance criteria for stability testing of adhesive patches for transfer of transdermal medications
 - b. TIs are used to improve control of Delivered Dose Uniformity (DDU) in Orally Inhaled and Nasal Drug Products (OINDP) that combine test for performance of delivery device and formulation^[3]
 - c. TIs are used to assure a certain proportion of the products covered within a pre-specified acceptance interval.^[4]
2. Drug – response analyses.^[5]
3. Characterization of Image Registration Performance.^[6]
4. Environmental monitoring and regulations.^[7]
5. Assessment of Food Safety.^[8]

The goal of this paper is to demonstrate the calculation of TIs for any application. The SAS MACRO written by the author can be easily adapted to address multiple research questions.

CALCULATION OF TI

Calculation of two-sided or one-sided TI depends on the research question. Two-sided TI will encompass p proportion of population measurements. One-sided TI_{Lower} guarantees that p proportion of the populations measurements will not fall below a lower limit, and one-sided TI_{Upper} guarantees that p proportion of the population measurements will not exceed an upper limit.

NORMALLY DISTRIBUTED POPULATION

(1) Two-sided TI (Howe, 1969)

TI = Mean +/- KFactor2s * Standard Deviation (SD)

$$KFactor2s = \sqrt{\frac{(n-1)(1 + \frac{1}{n})Z_{(1+p)/2}^2}{\chi_{(a,n-1)}^2}}$$

TI_{Lower} = Mean - KFactor2s*SD;

TI_{Upper} = Mean + KFactor2s*SD;

(2) One-sided TI (Natrella, 1963)

TI_{Upper} = Mean + KFactor1s * Standard Deviation

TI_{Lower} = Mean - KFactor1s * Standard Deviation

$$A = 1 - \frac{Z_{(1-\alpha)}^2}{2(n-1)} \quad B = Z_p^2 - \frac{Z_{1-\alpha}^2}{n} \quad KFactor1s = \frac{Z_p + \sqrt{Z_p^2 - A*B}}{A}$$

NON-NORMAL DATA

The test for normality can be performed in SAS by PROC UNIVARIATE. It is important to remember that TIs are sensitive to normality. Non-parametric (NP) approach is recommended when p ≥ 90% and level of confidence is high. Please note, a sample of data is required to perform calculation of NP TI, and the theory developed by Turkey (1947) can be applied.

1. Sort data from lowest to highest.
2. **Two-sided NP TI** will be between the KthLargest and (n - KthLargest + 1) values of this data.
 - a) KthLargest is the largest integer k that satisfies the equation.

$$CDF \text{ Beta}_{n-2k+1,2k}(p) \leq \alpha$$

where CDF Beta is a cumulative distribution function of the beta distribution with the parameters: n - 2k+1 and 2k.

b) TI_{Lower} = KthLargest value; and TI_{Upper} = n - KthLargest + 1 value.

3. One-sided NP TI

a) KthLargest is the largest integer k that satisfies the equation: $CDF \text{ Beta}_{n-k+1,k}(p) \leq \alpha$.

TI_{Upper} = n - KthLargest; and TI_{Lower} = KthLargest value.

DATABASE

The U.S. National Health and Nutrition Examination Survey (NHANES) is a representative sample of non-institutionalized US civilian residents. Descriptive statistics for NHANES 2013-2014 shows that the mean of Systolic Blood Pressure (SBP) for adult population 60 or greater years of age equals to 133.46 (95% CI: 132.51-134.41), Table 1.

Table 1: Descriptive statistics for Systolic Blood Pressure for American Adults NHANES 2013-2014

Age (years)	N	Mean	Std Dev	Lower 95% CL for Mean	Upper 95% CL for Mean	Std Error	Minimum	Maximum
>= 60	1713	133.46	20.00	132.51	134.41	0.48	70.67	228.67

Hypertension is a major risk factor for cardiovascular disease and stroke that are two leading causes of death in the US population.^{9,10} In 2014, the Eighth Report of the Joint National Committee on Prevention, Detection, Evaluation, and Treatment of High Blood Pressure (JNC 8) recommended to start treating

patients with Systolic (SBP) or Diastolic (DBP) blood pressure when the level is up or above 150mmHg and/or 90mmHg respectively for people ≥ 60 years of age.^[10,11] However, in 2017, the new guidance recommended to treat everyone with SBP/DBP $\geq 130/90$.^[12]

EXAMPLES

EXAMPLE #1: The objective is to find the proportion of American population ≥ 60 years of age that will require treatment for high SBP following recommendation from guidance JNC 8 in 2014, and from JNC in 2017.

A one-sided Lower Tolerance Limit will be a value of SBP with a guarantee that measurements in p-proportion of the American population 60 years of age or older will not fall below (will be equal or exceed) with $(1-\alpha)$ level of confidence. If $Tl1s_{Lower}$ reaches a threshold 150 mmHg (as from 2014 guidance), or 130 mmHg (as from 2017 guidance), than that proportion of US population will require antihypertensive treatment with $(1-\alpha)$ probability.

SAS MACRO %TI from REFERENCE 13 was used to calculate parametric $Tl1s_{Lower}$ limit as following:

```
%RUN_TI(inds=e14, outds=final1, param=sbp, p1=20, p2=60, byp=5, alpha=1 );
proc print; format t1slower 8.; vars p gamma t1slower; run;
```

Output 1: Parametric Lower Tolerance Limits

P %	Confidence %	$Tl1s_{Lower}$
20	99	149
25	99	146
30	99	143
....		
50	99	132
55	99	130
60	99	127

CONCLUSIONS EXAMPLE #1

1. Assuming that the data follow the normal distribution, there is 99% confidence that SBP measurements in 20% of population of interest will not fall below 149mmHg (one-sided Tl_{lower} is equal to 149 for proportion $p=20\%$, and confidence $(1-\alpha) = 99\%$).

Assuming that the data follow the normal distribution, there is 99% confidence that at least 20% of the American population age ≥ 60 years will require antihypertensive medications, following recommendation from the guidance JNC 8 (2014) with a threshold to treat elderly patients at level of SBP ≥ 150 mmHg.

2. Assuming that the data follow the normal distribution, there is 99% confidence that SBP measurements in 55% of population of interest will not fall below 130mmHg (one-sided Tl_{lower} is equal to 130 for proportion $p=55\%$, and confidence $(1-\alpha) = 99\%$).

Assuming that the data follow the normal distribution, there is 99% confidence that at least 55% of the American population age ≥ 60 years will require antihypertensive medications, following recommendation from the guidance JNC-2017 with a threshold to treat all patients at level of SBP ≥ 130 mmHg.

EXAMPLE № 2: The objective is to find an interval that covers individual values of SBP for at least 80% of American population ≥ 60 years of age with 95%, and 99% of confidence.

Calculation of the two-sided NP Tolerance Intervals will be the intervals that cover individual values of SBP for at least 80% of population of interest with 95%, and 99% of confidence. The nonparametric (distribution-free) TIs make no special distributional assumption.

SAS MACRO %NPTI from REFERENCE 14 was used to calculate NPTI as following:

```
%RUN_NPTI(inds=e14, outds=final1, param=sbp, p1=80, p2=80, byp=5, alpha=5);
%RUN_NPTI(inds=e14, outds=final2, param=sbp, p1=80, p2=80, byp=5, alpha=1);
DATA FINAL; SET FINAL1-FINAL2; RUN;

PROC PRINT DATA=FINAL;
FORMAT TILOWER TIUPPER 8.;
VAR P GAMMA TILOWER TIUPPER;
RUN;
```

Output 2: Non-parametric Two-sided Tolerance Intervals

P %	Confidence %	TI _{Lower}	TI _{Upper}
80	95	102	160
80	99	101	161

CONCLUSIONS EXAMPLE #2

1. There is 95% confidence that at least 80% of the American population age ≥ 60 years will have SBP measurements between 102 and 160 mmHg (two-sided TI is equal to 102-160 for proportion $p=80\%$, and confidence $(1-\alpha) = 95\%$).
2. There is 99% confidence that at least 80% of the American population age ≥ 60 years will have SBP measurements between 101 and 161 mmHg (two-sided TI is equal to 101-161 for proportion $p=80\%$, and confidence $(1-\alpha) = 99\%$).

CONCLUSION

The goal of TI is population predictive inferences that can be widely used in clinical research. However, the use of TI is underestimated perhaps due to the complex calculations. This paper explains the formulas that can be used by any programmer. The author provided a user-friendly SAS MACRO that was developed with use of SAS/BASE. This MACRO can be easily adapted for different research questions.

REFERENCES

1. U.S. Department of Health and Human Services Food and Drug Administration Center for Drug Evaluation and Research (CDER) Center for Biologics Evaluation and Research (CBER); Pharmaceutical Quality/CMC Current Good Manufacturing Practices (CGMPs) "Submission of Quality Metrics Data" Guidance for Industry Revision 1 November 2016
2. Thomas A. Little, 2016; Essentials in Tolerance Design and Setting Specification Limits"; http://thomasalittleconsulting.com/publications/articles/Essentials_in_Tolerance_Design_and_Setting_Specification_Limits.pdf;
3. Tsong Y, Dong X, Shen M, Lostritto Rt; 2015; "Quality Assurance Test Of Delivered Dose Uniformity Of Multiple-Dose Inhaler And Dry Powder Inhaler Drug Products" ; J Biopharm Stat. 2015;25(2):328-38. doi: 10.1080/10543406.2014.972510.

4. Dong X, Tsong Y, Shen M, Zhong J.; 2015; "Using Tolerance Intervals for Assessment Of Pharmaceutical Quality". *J Biopharm Stat.* 2015; 25(2):317-27. doi: 10.1080/10543406.2014.972512
5. Jing Pan; "The Application of Tolerance Interval in Defining Drug Response for Biomarker"; *PharmaSUG China 2015 - Paper 57*
6. Andriy Fedorov, William M. Wells, Ron Kikinis, Clare M. Tempny, Mark G. Vangel; 2014 "Application of Tolerance Limits to the Characterization of Image Registration Performance"; *IEEE Trans Med Imaging.* 2014 July ; 33(7): 1541–1550. doi:10.1109/TMI.2014.2317796.
7. Robert W. Smith; 2002; The Use of Random-Model Tolerance Intervals in Environmental Monitoring and Regulation; *Journal of Agricultural, Biological, and Environmental Statistics*;
8. Chen H, Kishino H, 2015; Hypothesis Testing of Inclusion of the Tolerance Interval for the Assessment of Food Safety. *PLoS ONE* 10(10): e0141117. <https://doi.org/10.1371/journal.pone.0141117>
9. Merai R, Siegel C, Rakotz M, Basch P, Wright J, Wong B; DHSc., Thorpe P. CDC Grand Rounds: A Public Health Approach to Detect and Control Hypertension. *MMWR Morb Mortal Wkly Rep.* 2016 Nov 18;65(45):1261-1264
10. Chobanian AV, Bakris GL, Black HR, Cushman WC, Green LA, Izzo JL Jr, et al.; 2003; The seventh report of the Joint National Committee on Prevention, Detection, Evaluation, and Treatment of High Blood Pressure: The JNC 7 report. *JAMA* 289(19):2560–72.
11. Paul A. James, MD; Suzanne Oparil, MD; Barry L. Carter, PharmD; William C. Cushman, MD; Cheryl Dennison-Himmelfarb, RN, ANP, PhD; Joel Handler, MD; Daniel T. Lackland, DrPH; Michael L. LeFevre, MD, MSPH; Thomas D. MacKenzie, MD, MSPH; Olugbenga Ogedegbe, MD, MPH, MS; Sidney C. Smith Jr, MD; Laura P. Svetkey, MD, MHS; Sandra J. Taler, MD; Raymond R. Townsend, MD; Jackson T. Wright Jr, MD, PhD; Andrew S. Narva, MD; Eduardo Ortiz, MD, MPH 2014 Evidence-Based Guideline for the Management of High Blood Pressure in Adults Report From the Panel Members Appointed to the Eighth Joint National Committee (JNC 8); *JAMA.* 2014; 311(5):507-520. doi:10.1001/jama.2013.284427; Published online December 18, 2013.
12. Whelton PK, Carey RM, Aronow WS, Casey Jr DE, Collins KJ, Dennison Himmelfarb C, DePalma SM, Gidding S, Jamerson KA, Jones DW, MacLaughlin EJ, Muntner P, Ovbiagele B, Smith Jr SC, Spencer CC, Stafford RS, Taler SJ, Thomas RJ, Williams Sr KA, Williamson JD, Wright Jr JT, 2017 ACC/AHA/AAPA/ABC/ACPM/AGS/APhA/ASH/ASPC/NMA/PCNA Guideline for the Prevention, Detection, Evaluation, and Management of High Blood Pressure in Adults, *Journal of the American College of Cardiology* (2017), doi: 10.1016/j.jacc.2017.11.006.

CONTACT INFORMATION

Marina Komaroff, Dr.P.H., M.P.H.

Director – Biometrics
Product Development/Clinical Operations/Regulatory
Noven Pharmaceuticals, Inc.
100 Town Square Place, 5th Floor
Jersey City, NJ 07310
Tel: 551-233-2645
Email: Mkomaroff@noven.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are registered trademarks or trademarks of their respective companies.

REFERENCE 13: PARAMETRIC TOLERANCE INTERVALS

```

** DESCRIPTION: call for macro ****;
%TI(num=      ,/*number of runs to keep output data sets with different names*/
      ds=      ,/*name for input data set                                     */
      param=   ,/*individual parameter of interest                         */
      p=       ,/*fixed proportion of population for TI to cover          */
      alpha=   ,/*to calculate the level of confidence gamma=1-alpha      */);

%MACRO TI(num= , ds= , param= , p= , alpha= );
TITLE "PARAMETRIC TOLERANCE INTERVALS";
proc means data=&ds noprint;
  var &param;
  output out=st mean=mean std=sd n=n;
run;
data _NULL_; set st;
  call symput("n", n );
  call symput("mean", mean );
  call symput("sd", sd );
run;

data TI2SD&num;
format TIUpper TI1Low TI1SLower TI1Supper 8.3;
mean=&mean; SD=&SD; n=&n;
p=&p/100;      * proportion ***;
alpha=&alpha/100;
gamma=1-alpha; * confidence ***;

* calculate K factor***;
z=probit((1+p)/2);
K1= Z *(sqrt(1 + (1/n)));

chisq=CINV(alpha, n-1);
K2=sqrt( (n-1) / chisq );

K=K1*K2;
TI1lower = mean - K*SD;
TI1upper = mean + K*SD;

* one-sided interval**;
Zp=probit(p);
Zgamma=probit(gamma);

A = 1 - ((Zgamma**2) / (2*(n-1)));
B = Zp**2 - (Zgamma**2)/n;
k1s=(Zp + SQRT(Zp**2 - A*B))/A;

TI1SLower = mean - k1s*SD;
TI1Supper = mean + k1s*SD;

output;
run;
%MEND;

```

REFERENCE 14: NON-PARAMETRIC TOLERANCE INTERVALS

```

** DESCRIPTION: call for macro ****;
%NPTI(num= , /*number of runs to keep output data sets with differ names*/
      ds= , /*name for input data set */
      param= , /*individual parameter of interest */
      p= , /*fixed proportion of values of interest for TI to cover */
      alpha= , /*to calculate level of confidence gamma=1-alpha */ );

%MACRO NPTI(num= , ds= , param= , p= , alpha= );
TITLE "NON-PARAMETRIC TOLERANCE INTERVALS";
%GLOBAL nobs; %let nobs=.;
%let dsnid = %sysfunc(open(&ds));
%if &dsnid %then %do; /* If the open was successful get the;
    %let nobs=%sysfunc(attrn(&dsnid,nlobs));
    %let rc =%sysfunc(close(&dsnid)); /* number of observations and CLOSE;
%end;
* sort data set by parameter ***;
proc sort data=&ds out=myds; by &param ; run;

* find k-factor and apply to find the upper and lower limits from the data;
data mydsk;
set myds;
retain kfactor kfactorls 0;
n=&nobs;
p=&p/100; /* proportion ***;
alpha=&alpha/100;
gamma=1-alpha; * confidence ***;

kfactor=KFactor + 1;
CDFBeta=probbeta(p, n-2*kfactor+1,2*kfactor);
if CDFBeta > .z then do;
    if CDFBeta le alpha then do; flag="L"; end;
    if CDFBeta gt alpha then do; flag="S"; end;
end;
kFactorls=KFactorls + 1;
CDFBetals=probbeta(p, n-kfactorls+1,kfactorls);
if CDFBetals > .z then do;
    if CDFBetals le alpha then do; flagls="L"; end;
    if CDFBetals gt alpha then do; flagls="S"; end;
end;
run;

proc sort data=mydsk; by kfactor flag; run;
data _NULL_; set mydsk;
by kfactor flag;
if last.flag and flag="L" then do;
    call symput("TIUpper", n - kfactor + 1);
    call symput("TILower", kfactor );
end;
run;

proc sort data=mydsk; by kfactorls flagls; run;
data _Null_; set mydsk;
by kfactorls flagls;
if last.flagls and flagls="L" then do;
    call symput("TIlsUpper", n - kfactorls );
    call symput("TIlsLower", kfactorls );

```



```

end;
run;
data NPTLI&num;
  set mydsk end=lastobs;
format TIUpper TILower TI1SLower TI1SUpper 8.3;
retain TI1sLower TI1sUpper TILower TIUpper 0;
keep n p alpha gamma KFactor KFactor1s TI1sLower TI1sUpper TILower TIUpper ;
  if _n_ = &TI1sUpper then TI1sUpper=&param;
  if _n_ = &TI1sLower then TI1sLower=&param;
  if _n_ = &TIUpper then TIUpper=&param;
  if _n_ = &TILower then TILower=&param;
if lastobs then output;
run;
%MEND;

** THE MACRO RUN_TI TO RUN MACRO NPTI MULTIPLE TIMES ****;
** DESCRIPTION: call for macro ****;
%RUN_NPTI(inds= , /*name for input data set */
  outds= , /*name of output data set */
  param= , /*individual parameter of interest */
  p1= , /*beginning proportion of values of interest for TI to cover*/
  p2= , /*end proportion of values of interest for TI to cover */
  byp= , /*increment for proportions */
  alpha= , /*to calculate level of confidence gamma=1-alpha */ );

%MACRO RUN_NPTI(inds= , outds= , param= , p1= , p2= , byp= , alpha= );
%let ni=0;
%DO pi=&p1 %TO &p2 %BY &byp;
  %let ni=%eval(&ni + 1);
  %NPTI(num=&ni, ds=&inds , param=&param , p=&pi, alpha=&alpha );
%END;
data &outds; set
  %DO i =1 %to &ni;
    NPTLI&i
  %end;
; run;
%MEND;

```